***********************************************************************************
***	Replication file for:                                                     	***
*** Braun, S. T. and Stuhler, J. (2024). The Economic Consequences of 			***
***	Being Widowed by War: A Life-Cycle Perspective.	     						***
***																				***
***	Journal of Public Economics                                                 ***
***   							                                                ***
*** Script:		_1-ghs-1-prep-life-cycle.do					   			 		***	
*** Purpose:	data preparation, life-cycle data GHS-2, combindes surveys		***
***				A and TA														***
***																				***
***********************************************************************************

***
* Note: Time variable t counts months since 01/1900
* Observation period: t=229...1068 referring to 01/1919-12/1988 
***


*** Append data

use "$widowsghs/processed/intermediate/LV2A_PANELDATEN_Index_edt.dta", clear
gen lv2 = "a"
append using "$widowsghs/processed/intermediate/LV2TA_PANELDATEN_Index_edt.dta"
replace lv2 = "t" if lv2 == ""


*** Generate variables on sector and occupation

*** Einteilung Branchen & Berufliche Stellung

/*
(01) LAND- UND FORSTWIRTSCHAFT, GÄRTNEREI
(02) FISCHEREI, FISCHZUCHT
(03) ENERGIEWIRTSCHAFT
(04) BERGBAU
(05) CHEMISCHE INDUSTRIE UND MINERALÖLVERARBEITUNG
(06) KUNSTSTOFFVERARBEITUNG
(07) STEINE UND ERDEN
(08) GIESSEREI, METALLVERARBEITUNG
(09) MASCHINENBAU
(10) ELEKTROTECHNIK
(11) HOLZ-, PAPIER- UND DRUCKGEWERBE
(12) LEDER, TEXTIL, BEKLEIDUNG
(13) NAHRUNGS- UND GENUSSMITTELGEWERBE
(14) BAUHAUPTGEWERBE
(15) BAUHILFSGEWERBE
(16) GROSSHANDEL
(17) HANDELSVERMITTLUNG
(18) EINZELHANDEL
(19) BUNDESBAHN, REICHSBAHN
(20) POST
(21) NACHRICHTEN, VERKEHR, TOURISTIK
(22) KREDITINSTITUTE (BANK)
(23) VERSICHERUNG
(24) DIENSTLEISTUNGEN
(25) ORGANISATIONEN OHNE ERWERBSCHARAKTER
(26) PRIVATE HAUSHALTE
(27) KÖRPERSCHAFTEN, BEHÖRDEN, VERTEIDIGUNG
(28) SOZIALVERSICHERUNG
(30) HANDEL, GEWERBE (MITHELFENDE)
(31) FREIE PRAXIS (MITHELFENDE)
*/

* Distinguish between three sectors

gen sector_3_t = .
replace sector_3_t = 1 if (branche_t >= 1 & branche_t <= 2)
replace sector_3_t = 2 if (branche_t >= 3 & branche_t <= 18)
replace sector_3_t = 3 if (branche_t >= 19 & branche_t <= 28)

label define sector_3_label 1 "Primary sector" 2 "Secondary sector" 3 "Tertiary sector"
label values sector_3_t sector_3_label

label var sector_3_t "Classification of employment at time t into three sectors"

* Distinguish between eight sectors

gen sector_8_t = .
replace sector_8_t = 1 if (branche_t >= 1 & branche_t <= 2)
replace sector_8_t = 2 if (branche_t >= 3 & branche_t <= 13)
replace sector_8_t = 3 if (branche_t == 14 | branche_t == 15)
replace sector_8_t = 4 if (branche_t >= 16 & branche_t <= 18)
replace sector_8_t = 5 if (branche_t >= 19 &  branche_t <= 21)
replace sector_8_t = 6 if (branche_t >= 22 &  branche_t <= 23)
replace sector_8_t = 7 if (branche_t == 24)
replace sector_8_t = 8 if (branche_t >= 25 & branche_t <= 28)

label define sector_8_label 1 "Agriculture" 2 "Industry" 3 "Construction" 4 "Trade" 5 "Transport and tourism" 6 "Financial services" 7 "Other private services" 8 "Public services and organisations"																																											
label values sector_8_t sector_8_label

label var sector_8_t "Classification of employment at time t into eight sectors"

drop branche_t


*** Berufliche Stellung

/*
(10) LANDWIRT UNTER 10 HA
(11) LANDWIRT 10 BIS 20 HA
(12) LANDWIRT 20 BIS 50 HA
(13) LANDWIRT 50 HA UND MEHR
(14) FREIBERUFLER (Z.B. MEDIZIN. HILFSBERUFE)
(15) AKADEMISCHER BERUF, 1 MITARBEITER ODER ALLEIN
(16) AKADEMISCHER BERUF, 2–9 MITARBEITER
(17) AKADEMISCHER BERUF, 10 UND MEHR MITARBEITER
(18) LANDWIRT MIT UNBEKANNTER NUTZFLÄCHE
(19) LANDWIRT OHNE NÄHERE ANGABEN
(20) HEIMARBEIT
(21) SELBSTÄNDIG, 1 MITARBEITER ODER ALLEIN
(22) SELBSTÄNDIG, 2–9 MITARBEITER
(23) SELBSTÄNDIG, 10–49 MITARBEITER
(24) SELBSTÄNDIG, 50 UND MEHR MITARBEITER
(28) SELBSTÄNDIG, NÄHERE DEFINITION NICHT GEWUSST
(29) SELBSTÄNDIG OHNE NÄHERE ANGABEN
(30) FAMILIENHILFE
(40) BEAMTE EINFACHER DIENST
(41) BEAMTE MITTLERER DIENST
(42) BEAMTE GEHOBENER DIENST
(43) BEAMTE HÖHERER DIENST
(48) BEAMTE MIT UNBEKANNTEM STATUS
(49) BEAMTE OHNE WEITERE ANGABEN
(50) WERKMEISTER
(51) ANGESTELLTE MIT EINFACHER TÄTIGKEIT
(52) ANGESTELLTE MIT SCHWERER TÄTIGKEIT
(53) ANGESTELLTE MIT VERANTWORTLICHER TÄTIGKEIT
(54) ANGESTELLTE MIT FÜHRENDER TÄTIGKEIT
(58) ANGESTELLTE MIT UNBEKANNTEM STATUS
(59) ANGESTELLTE OHNE WEITERE ANGABEN
(60) UNGELERNTE ARBEITER
(61) ANGELERNTE ARBEITER
(62) FACHARBEITER
(63) VORARBEITER
(64) MEISTER
(68) ARBEITER MIT UNBEKANNTEM STATUS
(69) ARBEITER OHNE WEITERE ANGABEN
(74) PRAKTIKANTEN/VOLONTÄRE
(75) PFLICHTJAHR
(76) HAUSHALTSJAHR/DIAKONISCHES JAHR
(82) AUSZUBILDENDE
(84) SOLDAT IM KRIEGSDIENST


*/

gen 	occupation_7_t = 1 if (beruf_stellung_t >= 10 & beruf_stellung_t <= 13)
replace occupation_7_t = 2 if (beruf_stellung_t >= 14 & beruf_stellung_t <= 17) | (beruf_stellung_t >= 21 & beruf_stellung_t <= 29)
replace occupation_7_t = 3 if (beruf_stellung_t == 30)
replace occupation_7_t = 4 if (beruf_stellung_t >= 40 & beruf_stellung_t <= 49)
replace occupation_7_t = 5 if (beruf_stellung_t >= 50 & beruf_stellung_t <= 59)
replace occupation_7_t = 6 if (beruf_stellung_t >= 60 & beruf_stellung_t <= 69)
replace occupation_7_t = 7 if (beruf_stellung_t >= 74 & beruf_stellung_t <= 82) | beruf_stellung_t == 84


label define occupation_7_label 1 "Farmer" 2 "Self-employed" 3 "Family co-worker" 4 "Civil servant" 5 "White collar worker" 6 "Blue collar worker" 7 "Apprentice or military service"
label values occupation_7_t occupation_7_label

label var occupation_7_t "Classification of occupational status at time t into 7 categories"


order sector_3_t sector_8_t occupation_7_t, after(beruf_stellung_t)

drop beruf_stellung_t

sort fid t 
compress
save "$widowsghs/processed/LVS2_panel_data_IP.dta", replace


*** EOF
